

Subroutine 1

Versione “semplice” che scambia gli elementi della diagonale con se stessi:

```
transpose: move    $t0,$a0        # t0 = puntatore elem diagonali
             move    $t1,$a1        # t1 = contatore ciclo esterno
             mul     $t7,$a1,4      # t7 = 4n (n word)
outerloop: move    $t2,$t0        # t2 = scorre riga
             move    $t3,$t0        # t3 = scorre colonna
             move    $t4,$t1        # t4 = contatore ciclo interno

innerloop: lw     $t5,($t2)        # Scambio (t2) con (t3)
             lw     $t6,($t3)
             sw     $t6,($t2)
             sw     $t5,($t3)
             addi   $t2,4
             add    $t3,$t3,$t7
             addi   $t4,-1
             bgtz  $t4,innerloop # while (t4>0)

             add    $t0,$t0,$t7    # prox diagonale +n+1
             addi   $t0,4
             addi   $t1,-1
             bgtz  $t1,outerloop
             jr     $ra
```

Versione più ottimizzata che scambia solo gli elementi non diagonali (+istruz -tempo):

```
transpose: move    $t0,$a0        # t0 = puntatore elem diagonali
             move    $t1,$a1        # t1 = contatore ciclo esterno
             mul     $t7,$a1,4      # t7 = 4n (n word)
outerloop: ble     $t1,1,exitouter # while(t1>1)
             add    $t2,$t0,4      # t2 = scorre riga =right(t0)
             add    $t3,$t0,$t7    # t3 = scorre colonna =down(t0)
             move    $t4,$t1        # t4 = contatore ciclo interno

innerloop: ble     $t4,1,exitinner
             lw     $t5,($t2)        # Scambio (t2) con (t3)
             lw     $t6,($t3)
             sw     $t6,($t2)
             sw     $t5,($t3)
             addi   $t2,4
             add    $t3,$t3,$t7
             addi   $t4,-1
             b      innerloop

exitinner: add    $t0,$t0,$t7    # prox diagonale +n+1
             addi   $t0,4
             addi   $t1,-1
             b      outerloop
exitouter: jr     $ra
```

Subroutine 2

```
checksim: move    $t0,$a0           # t0 = puntatore elem diagonali
            move    $v0,$a1         # v0 = contatore ciclo esterno
            mul     $t7,$a1,4       # t7 = 4n (n word)
outerloop: ble     $v0,1,exitouter  # while(v0>1)
            add     $t2,$t0,4       # t2 = scorre riga =right(t0)
            add     $t3,$t0,$t7     # t3 = scorre colonna =down(t0)
            move    $t4,$v0        # t4 = contatore ciclo interno

innerloop: ble     $t4,1,exitinner
            lw      $t5,($t2)       # Abort se (t2) ≠ (t3)
            lw      $t6,($t3)
            bne     $t6,$t5,exitouter
            addi    $t2,4
            add     $t3,$t3,$t7
            addi    $t4,-1
            b       innerloop

exitinner: add     $t0,$t0,$t7     # prox diagonale +n+1
            addi    $t0,4
            addi    $v0,-1
            b       outerloop

exitouter: addi    $v0,-1         # se è simm v0=1 altrimenti v0>1
            jr      $ra           # quindi QUI v0=0 se simm o v0>0
```